

# Mathematical Programming: Modelling and Applications

Sonia Cafieri

LIX, École Polytechnique

`caferi@lix.polytechnique.fr`

October 2009

# Densest Subgraph problem: AMPL model

```
param n >= 1, integer;
set V := 1..n;
set E within {V,V};

# arc colours
param kmax default 10; # max number of colours
param k <= kmax, >= 0, integer, default 1;
param mu{E} >=0, integer, <= kmax;

# variables
var x{V} binary;
var y{(u,v) in E} >= 0, <= min(max(0, mu[u,v]-k+1), max(0,k-mu[u,v]+1));

# model
maximize densesubgraph : sum{(u,v) in E} y[u,v] - sum{v in V} x[v];

# linearization constraints
subject to lin1 {(u,v) in E} : y[u,v] <= x[u];
subject to lin2 {(u,v) in E} : y[u,v] <= x[v];
subject to lin3 {(u,v) in E} : y[u,v] >= x[u] + x[v] - 1;
```

# Densest Subgraph problem: AMPL data

```
param n := 15; # number of vertices
param k := 2; # color
```

```
param : E : mu :=
```

```
1 15 1
2 15 1
2 3 1
2 4 1
3 5 1
4 5 1
5 6 1
5 11 2
5 12 2
5 13 2
5 14 2
6 9 1
7 8 1
7 11 2
7 12 2
7 13 2
7 14 2
7 15 1
8 10 1
8 14 2
11 12 2
11 13 2
12 13 2
;
```

# Densest Subgraph problem: AMPL run

```
model densest_subgraph.mod;  
data densest_subgraph.dat;  
  
option solver cplex;  
solve;  
  
display x;
```